Mathematica Notebook for "Asymptotic Methods for Asset Market Equilibrium Analysis" by Judd and Guu

The following *Mathematica* program computes asset market equilibrium for two investors, one safe asset, and one or two risky assets.

The first command turns off annoying spelling queries.

```
In[1]:= Off[General::spell1]
```

■ The initial setup

■ Economic Model

Define the two risky assets' returns, Z and Y, in terms of zero-mean random variables z and y. ϵ is the scaling parameter, equal to the standard deviation in the ϵ-economy. Z is an asset with mean return R and variance ϵ^2, and is in positive net supply. It represents equity. To keep the formulas simple, we assume that the return on bonds, R, is 1. This is equivalent to assuming that the bond in the second period is the numeraire.

These facts imply that Z can be decomposed in the following manner:

```
In[2]:= Z = 1 + ϵ z;
```

Y is the derivative asset and has zero net supply. It has mean return $\epsilon\ \mu Y$, is partially correlated with Z through the $\epsilon\,\alpha\,z$ term, and has an orthogonal component $\epsilon\,y$. These facts imply that Y can be represented as

```
In[3]:= Y = ϵ μY  + ϵ α z + ϵ y;
```

The price of Z (Y) is p (q). We parametrize them in terms of the scaling parameter ϵ, the mean of Y, μY, and the premia of Z and Y, denoted Π and Ψ:

```
In[4]:= p = 1 - ϵ^2 Π ; q = ϵ μY - ϵ^2 Ψ ;
```

Note: The notation in the notebook corresponds to the notation in the paper except for a minor change. In *Mathematica*, the letter π is reserved for 3.14159... and so we could not use it as the risk premium for Z. Therefore, we use Π instead. To maintain the symmetry we also let Ψ denote the premium for Y.

W_i is final wealth of type i investor. θ is type 1 demand for equity. θe_i is type i endowment of equity and Be_i is type i endowment of bonds. ϕ is type 1's demand for Y; - ϕ is type 2 demand. In equilibrium, type 2 investors will hold

$\theta e_1 + \theta e_2 - \theta$ shares of stock. (*Mathematica* prevented us from using θ_i^e (B_i^e) to denote type *i* endowment of the risky asset (bond).)

```
In[5]:= θe₁ = θe₁; θe₂ = θe₂;

In[6]:= W₁ = (Be₁ + p θe₁ - p θ - q ϕ) + θ * Z + ϕ Y
        W₂ = (Be₂ + p θe₂ - p (θe₁ + θe₂ - θ) - q (-ϕ)) + (θe₁ + θe₂ - θ) * Z + (-ϕ) Y;
        W₁ = Expand[W₁];
        W₂ = Expand[W₂];

Out[6]= (1 + z ϵ) θ - θ (1 - ϵ² Π) + (y ϵ + z α ϵ + ϵ μY) ϕ - ϕ (ϵ μY - ϵ² Ψ) + Be₁ + (1 - ϵ² Π) θe₁
```

We compute the four first-order conditions for the two investors and the two risky assets. The bond demand is determined by substracting risky asset demand from initial wealth. We divide each first-order condition by ϵ to eliminate one degree of degeneracy. FOCij is the first-order condition of type *i* investor with respect to risky asset *j*, where $j = 1$ is equity and $j = 2$ refers to the derivative.

```
In[10]:= FOC11 = (1 / ϵ) D[u₁[W₁], θ] // Simplify
         FOC12 = (1 / ϵ) D[u₁[W₁], ϕ] // Simplify
         FOC21 = (1 / ϵ) D[u₂[W₂], θ] // Simplify
         FOC22 = (1 / ϵ) D[u₂[W₂], ϕ] // Simplify

Out[10]= (z + ϵ Π) u₁′[ϵ (z θ + ϵ θ Π + y ϕ + z α ϕ + ϵ ϕ Ψ) + Be₁ + (1 - ϵ² Π) θe₁]

Out[11]= (y + z α + ϵ Ψ) u₁′[ϵ (z θ + ϵ θ Π + y ϕ + z α ϕ + ϵ ϕ Ψ) + Be₁ + (1 - ϵ² Π) θe₁]

Out[12]= -(z + ϵ Π) u₂′[-z ϵ θ - ϵ² θ Π - y ϵ ϕ - z α ϵ ϕ - ϵ² ϕ Ψ + Be₂ + ϵ (z + ϵ Π) θe₁ + θe₂ + z ϵ θe₂]

Out[13]= -(y + z α + ϵ Ψ) u₂′[-z ϵ θ - ϵ² θ Π - y ϵ ϕ - z α ϵ ϕ - ϵ² ϕ Ψ + Be₂ + ϵ (z + ϵ Π) θe₁ + θe₂ + z ϵ θe₂]
```

Even though we have divided by ϵ, these first-order conditions describe the situation at ϵ=0.
These four first-order conditions define equilibrium. Define *G* to be the vector of first-order conditions.

■ Moment substitutions and other substitutions

We next define lists of substitutions that allow us to compute the moments of *z* and *y*. The doubly subscripted term $\mu_{m,n}$ denotes the expectation of $z^m\ y^n$. We often need to compute expectations of our Taylor series expansions. We could invoke *Mathematica*'s Integrate command, but we can do better by searching for specific terms in our polynomials and replacing them with their integrals. For example, whenever we see an isolated *z* in an expression we know that its integral just replaces the *z* term by the mean of *z*, which is denoted as $\mu_{1,0}$. This works better than integration in *Mathematica* since we know that our function is a polynomial whereas *Mathematica* needs to spend time to ascertain that fact. The following replacement rules implement this approach to integration.

```
In[14]:= ownmoments = {z → μ1,0, z^m_ → μm,0, y → μ0,1, y^m_ → μ0,m}
         crossmoments = {z^m_ y^n_ → μm,n, z y^n_ → μ1,n, z^m_ y → μm,1}

Out[14]= {z → μ1,0, z^m_ → μm,0, y → μ0,1, y^m_ → μ0,m}

Out[15]= {y^n_ z^m_ → μm,n, y^n_ z → μ1,n, y z^m_ → μm,1}
```

Our definition of Y decomposed it into its mean, it covariance with Z, and the orthogonal residual. Without loss of generality, we set the mean of z and y to zero.

```
In[16]:= zeromean = {μ0,1 → 0, μ1,0 → 0};
```

We make z and y orthogonal and set the variance of z equal to 1.

```
In[17]:= orthogonal = {μ1,1 → 0, μ2,0 → 1};
```

We now gather all the substitions into one list.

```
In[18]:= moments = Join[ownmoments, crossmoments, zeromean, orthogonal]

Out[18]= {z → μ1,0, z^m– → μm,0, y → μ0,1, y^m– → μ0,m, y^n– z^m– → μm,n,
          y^n– z → μ1,n, y z^m– → μm,1, μ0,1 → 0, μ1,0 → 0, μ1,1 → 0, μ2,0 → 1}
```

The following substitutions denote the case where there is no derivative security. They say that all cross-moments between z and y are zero. Note that y must be nonzero since we set its variance to 1. However, if all of its moments are uncorrelated with all moments of the endowment then it is a pure gamble and will not be traded in equilibrium.

```
In[19]:= NoDeriv = Table[μm,n → 0, {m, 1, 6}, {n, 1, 6}] // Flatten

Out[19]= {μ1,1 → 0, μ1,2 → 0, μ1,3 → 0, μ1,4 → 0, μ1,5 → 0, μ1,6 → 0, μ2,1 → 0, μ2,2 → 0, μ2,3 → 0,
          μ2,4 → 0, μ2,5 → 0, μ2,6 → 0, μ3,1 → 0, μ3,2 → 0, μ3,3 → 0, μ3,4 → 0, μ3,5 → 0, μ3,6 → 0,
          μ4,1 → 0, μ4,2 → 0, μ4,3 → 0, μ4,4 → 0, μ4,5 → 0, μ4,6 → 0, μ5,1 → 0, μ5,2 → 0, μ5,3 → 0,
          μ5,4 → 0, μ5,5 → 0, μ5,6 → 0, μ6,1 → 0, μ6,2 → 0, μ6,3 → 0, μ6,4 → 0, μ6,5 → 0, μ6,6 → 0}
```

We define a convenient list of simplifying substitutions which *Mathematica* did not automatically execute.

```
In[20]:= sub1 = {1/2 (2 Be1 + 2 θe1) → (Be1 + θe1), 1/2 (2 Be2 + 2 θe2) → (Be2 + θe2)}

Out[20]= {1/2 (2 Be1 + 2 θe1) → Be1 + θe1, 1/2 (2 Be2 + 2 θe2) → Be2 + θe2}
```

■ Solvability Condition in Theorem 4

We check for the nonsingularity of the solvability matrix, $G_{\epsilon,\Lambda}$ where Λ=(θ,ϕ,Π,Ψ). First define the solvability matrix:

```
In[21]:= Gϵ,Λ =
          {{D[FOC11, {ϵ, 1}, θ], D[FOC11, {ϵ, 1}, ϕ], D[FOC11, {ϵ, 1}, Ψ], D[FOC11, {ϵ, 1}, Π]},
           {D[FOC12, {ϵ, 1}, θ], D[FOC12, {ϵ, 1}, ϕ], D[FOC12, {ϵ, 1}, Ψ], D[FOC12, {ϵ, 1}, Π]},
           {D[FOC21, {ϵ, 1}, θ], D[FOC21, {ϵ, 1}, ϕ], D[FOC21, {ϵ, 1}, Ψ], D[FOC21, {ϵ, 1}, Π]},
           {D[FOC22, {ϵ, 1}, θ], D[FOC22, {ϵ, 1}, ϕ], D[FOC22, {ϵ, 1}, Ψ], D[FOC22, {ϵ, 1}, Π]}};
```

Evaluate $G_{\epsilon,\Lambda}$ at ϵ=0 and display it in matrix form:

```
In[22]:= G_{ε,Λ} = G_{ε,Λ} /. {ε -> 0};
         G_{ε,Λ} // MatrixForm
```

Out[23]//MatrixForm=

$$\begin{pmatrix} z^2\, u_1''[Be_1+\theta e_1] & z\,(y+z\,\alpha)\, u_1''[Be_1+\theta e_1] & 0 & u_1'[Be_1+\theta e_1] \\ z\,(y+z\,\alpha)\, u_1''[Be_1+\theta e_1] & (y+z\,\alpha)^2\, u_1''[Be_1+\theta e_1] & u_1'[Be_1+\theta e_1] & 0 \\ z^2\, u_2''[Be_2+\theta e_2] & -z\,(-y-z\,\alpha)\, u_2''[Be_2+\theta e_2] & 0 & -u_2'[Be_2+\theta e_2] \\ z\,(y+z\,\alpha)\, u_2''[Be_2+\theta e_2] & -(-y-z\,\alpha)\,(y+z\,\alpha)\, u_2''[Be_2+\theta e_2] & -u_2'[Be_2+\theta e_2] & 0 \end{pmatrix}$$

Integrate the elements of $G_{\epsilon,\Lambda}$ by replacing powers of z and y with their moments.

```
In[24]:= ExpG = Expand[G_{ε,Λ}] //. moments;
         ExpG // MatrixForm
```

Out[25]//MatrixForm=

$$\begin{pmatrix} u_1''[Be_1+\theta e_1] & \alpha\, u_1''[Be_1+\theta e_1] & 0 & u_1'[Be_1+\theta e_1] \\ \alpha\, u_1''[Be_1+\theta e_1] & \alpha^2\, u_1''[Be_1+\theta e_1] + \mu_{0,2}\, u_1''[Be_1+\theta e_1] & u_1'[Be_1+\theta e_1] & 0 \\ u_2''[Be_2+\theta e_2] & \alpha\, u_2''[Be_2+\theta e_2] & 0 & -u_2'[Be_2+\theta e_2] \\ \alpha\, u_2''[Be_2+\theta e_2] & \alpha^2\, u_2''[Be_2+\theta e_2] + \mu_{0,2}\, u_2''[Be_2+\theta e_2] & -u_2'[Be_2+\theta e_2] & 0 \end{pmatrix}$$

Compute the determinant of $E\{G_{\epsilon,\Lambda}\}$.

```
In[26]:= (Det[ExpG] // Simplify) //. moments
```

Out[26]= $-\mu_{0,2}\,(u_2'[Be_2+\theta e_2]\, u_1''[Be_1+\theta e_1] + u_1'[Be_1+\theta e_1]\, u_2''[Be_2+\theta e_2])^2$

As long as $\mu_{0,2}$ is not zero, this determinant is strictly positive and $G_{\epsilon,\Lambda}$ is nonsingular

■ Bifurcation equations: series construction

We construct substitutions that will define the equilibrium mappings from the scaling parameter ϵ to the equilibrium values of (θ,ϕ,Π,Ψ).

```
In[27]:= PortfolioSubs = {θ → θ[ε], ϕ → ϕ[ε]}
```

Out[27]= $\{\theta \to \theta[\epsilon],\ \phi \to \phi[\epsilon]\}$

```
In[28]:= PremiaSubs = {Π → Π[ε], Ψ → Ψ[ε]}
```

Out[28]= $\{\Pi \to \Pi[\epsilon],\ \Psi \to \Psi[\epsilon]\}$

To construct the equilibrium equations we take the first-order conditions and replace the the portfolio variables, θ and ϕ, and the risk premium variables, Π and Ψ, with their equilibrium maps.

In[29]:=
```
EQM11 = FOC11 /. PortfolioSubs /. PremiaSubs
EQM21 = FOC21 /. PortfolioSubs /. PremiaSubs
EQM12 = FOC12 /. PortfolioSubs /. PremiaSubs
EQM22 = FOC22 /. PortfolioSubs /. PremiaSubs
```

Out[29]= $(z+\epsilon\,\Pi[\epsilon])\,u_1'[Be_1+\theta e_1\,(1-\epsilon^2\,\Pi[\epsilon])+\epsilon\,(z\,\theta[\epsilon]+\epsilon\,\theta[\epsilon]\,\Pi[\epsilon]+y\,\phi[\epsilon]+z\,\alpha\,\phi[\epsilon]+\epsilon\,\phi[\epsilon]\,\Psi[\epsilon])]$

Out[30]= $-(z+\epsilon\,\Pi[\epsilon])\,u_2'[Be_2+\theta e_2+z\,\epsilon\,\theta e_2-z\,\epsilon\,\theta[\epsilon]-\epsilon^2\,\theta[\epsilon]\,\Pi[\epsilon]+\epsilon\,\theta e_1\,(z+\epsilon\,\Pi[\epsilon])-y\,\epsilon\,\phi[\epsilon]-z\,\alpha\,\epsilon\,\phi[\epsilon]-\epsilon^2\,\phi[\epsilon]\,\Psi[\epsilon]]$

Out[31]= $(y+z\,\alpha+\epsilon\,\Psi[\epsilon])\,u_1'[Be_1+\theta e_1\,(1-\epsilon^2\,\Pi[\epsilon])+\epsilon\,(z\,\theta[\epsilon]+\epsilon\,\theta[\epsilon]\,\Pi[\epsilon]+y\,\phi[\epsilon]+z\,\alpha\,\phi[\epsilon]+\epsilon\,\phi[\epsilon]\,\Psi[\epsilon])]$

Out[32]= $-(y+z\,\alpha+\epsilon\,\Psi[\epsilon])\,u_2'[Be_2+\theta e_2+z\,\epsilon\,\theta e_2-z\,\epsilon\,\theta[\epsilon]-\epsilon^2\,\theta[\epsilon]\,\Pi[\epsilon]+\epsilon\,\theta e_1\,(z+\epsilon\,\Pi[\epsilon])-y\,\epsilon\,\phi[\epsilon]-z\,\alpha\,\epsilon\,\phi[\epsilon]-\epsilon^2\,\phi[\epsilon]\,\Psi[\epsilon]]$

The equilibrium equations are really the expectations 0=E{EQMij}. We will execute the integrations later, after the Taylor series expansions, since it is permissible to interchange the integration and differentiation operations.

We define abbrevations for the derivatives; this helps make the formulas more understandable than the ones automatically generated by *Mathematica.*

In[33]:=
```
UtilDerivs = {Derivative[n_][ui_][Bei_ + θei_] -> Ai[n]}
```

Out[33]= $\{u_{i\_}^{(n\_)}[Be_{i\_}+\theta e_{i\_}]\to A_i[n]\}$

We want to replace derivatives of the utility function, the $A_i[m]$ terms, with risk tolerance, τ_i, skew tolerance, ρ_i, and kurtosis tolerance, κ_i. The next substitution rule does that.

In[34]:= $\text{UtilParams} = \left\{A_{i\_}[2]\to-\frac{A_i[1]}{\tau_i},\ A_{i\_}[3]\to\frac{2\,\rho_i\,A_i[1]}{\tau_i{}^2},\ A_{i\_}[4]\to\frac{3\,\kappa_i\,A_i[1]}{\tau_i{}^3},\ A_{i\_}[1]\to 1\right\}$

Out[34]= $\left\{A_{i\_}[2]\to-\frac{A_i[1]}{\tau_i},\ A_{i\_}[3]\to\frac{2\,\rho_i\,A_i[1]}{\tau_i^2},\ A_{i\_}[4]\to\frac{3\,\kappa_i\,A_i[1]}{\tau_i^3},\ A_{i\_}[1]\to 1\right\}$

It is often useful to refer to the social risk tolerance, T, and express type i risk tolerance as a share, ν_i, of total risk tolerance. The following replacement rule allows us to do that.

In[35]:=
```
taureps = {τ1 + τ2 → T, τi_ → υi T, υ1 + υ2 → 1};
```

We now compute the power series expansions of the four equilibrium conditions. We fully expand each series so that the various powers of z and y are gathered together. This is necessary for our integration approach to work.

In[36]:=
```
EqmPow11 = (Normal[Series[EQM11 , {ϵ, 0, 6}]] /. sub1) //. UtilDerivs;
EqmPow11 = Expand[EqmPow11];
EqmPow12 = (Normal[Series[EQM12 , {ϵ, 0, 6}]] /. sub1) //. UtilDerivs;
EqmPow12 = Expand[EqmPow12];
EqmPow21 = (Normal[Series[EQM21 , {ϵ, 0, 6}]] /. sub1) //. UtilDerivs;
EqmPow21 = Expand[EqmPow21];
EqmPow22 = (Normal[Series[EQM22 , {ϵ, 0, 6}]] /. sub1) //. UtilDerivs;
EqmPow22 = Expand[EqmPow22];
```

We now take the expectation of each equilibrium equation's power series. Since z and y are the only random variables, we replace each power of z and y and each crossproduct by the appropriate moment.

```
In[44]:= EqmPow11 = EqmPow11 //. moments;
         EqmPow12 = EqmPow12 //. moments;
         EqmPow21 = EqmPow21 //. moments;
         EqmPow22 = EqmPow22 //. moments;
```

The final step in constructing the equilibrium expressions is to collect terms of like powers of ϵ and list the coefficients of each power of ϵ. This puts the equations in the proper arrangement for solving the problem.

```
In[48]:= Eqns11 = CoefficientList[EqmPow11, ϵ];
         Eqns12 = CoefficientList[EqmPow12, ϵ];
         Eqns21 = CoefficientList[EqmPow21, ϵ];
         Eqns22 = CoefficientList[EqmPow22, ϵ];
```

■ Solve the individual equations in sequence

■ Equation 1:

The coefficients of the ϵ^0 components of the equilibrium power series should be zero. We now check what they are.

```
In[52]:= {Eqns11[[1]], Eqns21[[1]], Eqns12[[1]], Eqns22[[1]]}

Out[52]= {0, 0, 0, 0}
```

This shows that we can continue. If this were not a vector of zeroes then we would know that our parameterization did not fulfill the necessary conditions of the Bifurcation Theorem.

■ Equation 2:

We compute the bifurcation point by choosing (θ[0],ϕ[0],Ψ[0],Π[0]) so that the ϵ components of the equilibrium equation expansion are zero. We first list the equations

```
In[53]:= Eq1 = Eqns11[[2]] // Simplify // Expand
         Eq2 = Eqns21[[2]] // Simplify // Expand
         Eq3 = Eqns12[[2]] // Simplify // Expand
         Eq4 = Eqns22[[2]] // Simplify // Expand
```

Out[53]= $\Pi[0]\, A_1[1] + \theta[0]\, A_1[2] + \alpha\, \phi[0]\, A_1[2]$

Out[54]= $-\Pi[0]\, A_2[1] - \theta e_1\, A_2[2] - \theta e_2\, A_2[2] + \theta[0]\, A_2[2] + \alpha\, \phi[0]\, A_2[2]$

Out[55]= $\Psi[0]\, A_1[1] + \alpha\, \theta[0]\, A_1[2] + \alpha^2\, \phi[0]\, A_1[2] + \mu_{0,2}\, \phi[0]\, A_1[2]$

Out[56]= $-\Psi[0]\, A_2[1] - \alpha\, \theta e_1\, A_2[2] - \alpha\, \theta e_2\, A_2[2] + \alpha\, \theta[0]\, A_2[2] + \alpha^2\, \phi[0]\, A_2[2] + \mu_{0,2}\, \phi[0]\, A_2[2]$

We next solve for bifurcation point.

In[57]:= **BifPt = Solve[{Eq1 == 0, Eq2 == 0, Eq3 == 0, Eq4 == 0}, {θ[0], φ[0], Ψ[0], Π[0]}][[1]]**

Out[57]= $\left\{\Psi[0] \to -\frac{\alpha A_1[2]\ (\theta e_1 A_2[2] + \theta e_2 A_2[2])}{A_1[2]\ A_2[1] + A_1[1]\ A_2[2]},\ \theta[0] \to \frac{A_1[1]\ (\theta e_1 A_2[2] + \theta e_2 A_2[2])}{A_1[2]\ A_2[1] + A_1[1]\ A_2[2]},\ \phi[0] \to 0,\ \Pi[0] \to -\frac{A_1[2]\ (\theta e_1 A_2[2] + \theta e_2 A_2[2])}{A_1[2]\ A_2[1] + A_1[1]\ A_2[2]}\right\}$

We simplify the expression, bringing together common factors.

In[58]:= **BifPt = BifPt // Simplify**

Out[58]= $\left\{\Psi[0] \to -\frac{\alpha\ (\theta e_1 + \theta e_2)\ A_1[2]\ A_2[2]}{A_1[2]\ A_2[1] + A_1[1]\ A_2[2]},\ \theta[0] \to \frac{(\theta e_1 + \theta e_2)\ A_1[1]\ A_2[2]}{A_1[2]\ A_2[1] + A_1[1]\ A_2[2]},\ \phi[0] \to 0,\ \Pi[0] \to -\frac{(\theta e_1 + \theta e_2)\ A_1[2]\ A_2[2]}{A_1[2]\ A_2[1] + A_1[1]\ A_2[2]}\right\}$

We next simplify by using the equilibrium conditions. Essentially, we want to replace all occurrences of $\theta e_1 + \theta e_2$ terms, the aggregate endowment of the risky asset, with its value, 1. In geneal, we let Θ be the total endowment of the risky asset; we set Θ=1. EqmSubs defines various substitutions that express the identity $\theta e_2 + \theta e_1 = \Theta$, andvec = - allow us to simplify various expressions.

In[59]:= **Θ = 1; EqmSubs = {θe₂ + θe₁ → Θ, θe₂ + θe₁ - Θ → 0, Θ - θe₂ - θe₁ → 0}**

Out[59]= $\{\theta e_1 + \theta e_2 \to 1,\ -1 + \theta e_1 + \theta e_2 \to 0,\ 1 - \theta e_1 - \theta e_2 \to 0\}$

Simplify BifPt by uskng the substitutions in EqmSubs

In[60]:= **BifPt = BifPt //. EqmSubs**

Out[60]= $\left\{\Psi[0] \to -\frac{\alpha A_1[2]\ A_2[2]}{A_1[2]\ A_2[1] + A_1[1]\ A_2[2]},\ \theta[0] \to \frac{A_1[1]\ A_2[2]}{A_1[2]\ A_2[1] + A_1[1]\ A_2[2]},\ \phi[0] \to 0,\ \Pi[0] \to -\frac{A_1[2]\ A_2[2]}{A_1[2]\ A_2[1] + A_1[1]\ A_2[2]}\right\}$

We do not like the utility derivatives, $A_i[j]$. We apply substitutions contained in UtilParams, defined above, that replace utility derivatives with indices such as risk tolerance.

In[61]:= **BifPt = BifPt //. UtilParams**

Out[61]= $\left\{\Psi[0] \to -\frac{\alpha}{\tau_1 \left(-\frac{1}{\tau_1} - \frac{1}{\tau_2}\right) \tau_2},\ \theta[0] \to -\frac{1}{\left(-\frac{1}{\tau_1} - \frac{1}{\tau_2}\right) \tau_2},\ \phi[0] \to 0,\ \Pi[0] \to -\frac{1}{\tau_1 \left(-\frac{1}{\tau_1} - \frac{1}{\tau_2}\right) \tau_2}\right\}$

We need to simplify this expression for BifPt. The Simplify command can handle this for us. (In general, one must be careful using the Simplify command since it can often take a long time to find the desired simplification or it will find a simplification other than the one you want.)

In[62]:= **BifPt = BifPt // Simplify**

Out[62]= $\left\{\Psi[0] \to \frac{\alpha}{\tau_1 + \tau_2},\ \theta[0] \to \frac{\tau_1}{\tau_1 + \tau_2},\ \phi[0] \to 0,\ \Pi[0] \to \frac{1}{\tau_1 + \tau_2}\right\}$

We now have an intuitive expression for the bifurcation point. The procedure was direct: solve for the bifurcation point in terms of utility derivatives and endowments, and then apply various simplifications to transform the solution into an expression involving elasticities and shares. Below we will apply similar sequences of simplifications without explanation.

The bifurcation point conforms to basic intutions. The risk premium, $\Pi[0]$, equals the inverse of the total risk tolerance, $\frac{1}{\tau_1+\tau_2}$. Type 1 holdings of equity equals it share of social risk tolerance. The risk premium of the derivative asset Y, $\Psi[0]$, equals $\frac{\alpha}{\tau_1+\tau_2}$ which is the product of the risk premium of equity and the covariance of Y with z. (This is equation (32) in the paper). The most surprising result is that the asymptotic holding of the derivative asset is zero. This does not mean that there is no demand for the derivative; it means that demand is smaller than order 0. We will see below that demand is of order ϵ.

■ Equation 3:

We next compute ($\theta'[0]$,$\phi'[0]$,$\Psi'[0]$,$\Pi'[0]$) by setting the ϵ^2 components of the equilibrium equations Taylor series expansions equal to zero. We immediately make the substitutions stated in UtilParams.

```
In[63]:= Eq1 = Eqns11[[3]] //. UtilParams // Expand;
         Eq2 = Eqns21[[3]] //. UtilParams // Expand;
         Eq3 = Eqns12[[3]] //. UtilParams // Expand;
         Eq4 = Eqns22[[3]] //. UtilParams // Expand;
```

Here is the first equation; the rest are similar.

```
In[67]:= Eq1
```

$$\text{Out[67]=}\ \frac{\rho_1\,\mu_{3,0}\,\theta[0]^2}{\tau_1^2}+\frac{2\,\rho_1\,\mu_{2,1}\,\theta[0]\,\phi[0]}{\tau_1^2}+\frac{2\,\alpha\,\rho_1\,\mu_{3,0}\,\theta[0]\,\phi[0]}{\tau_1^2}+\frac{\rho_1\,\mu_{1,2}\,\phi[0]^2}{\tau_1^2}+\frac{2\,\alpha\,\rho_1\,\mu_{2,1}\,\phi[0]^2}{\tau_1^2}+\frac{\alpha^2\,\rho_1\,\mu_{3,0}\,\phi[0]^2}{\tau_1^2}-\frac{\theta'[0]}{\tau_1}+\Pi'[0]-\frac{\alpha\,\phi'[0]}{\tau_1}$$

We now solve for the unknowns, $\theta'[0]$, $\phi'[0]$, $\Psi'[0]$, and $\Pi'[0]$. We suppress printing the first set of results since they take up too much space. The substitutions we make below will produce comprehensible expressions.

```
In[68]:= Sol2 = Solve[{Eq1 == 0, Eq2 == 0, Eq3 == 0, Eq4 == 0}, {θ'[0], ϕ'[0], Ψ'[0], Π'[0]}][[1]];
```

The solution involves , $\theta[0]$, $\phi[0]$, $\Psi[0]$, and $\Pi[0]$. We now replace these terms with the values found in our bifurcation point solution, BifPt.

```
In[69]:= Sol2 = Sol2 /. BifPt;
```

Note: we could have made these substitutions when we defined Eq1, Eq2, etc, but that would not be wise. If we had done that then the expressions in Eq1, Eq2, etc., would have been larger, making more work for *Mathematica*. This would not have been a major problem for this set of equations, but becomes important when we move to higher-order terms where the solutions become more complex.

We next simplify the result which produces a fairly compact form.

In[70]:= **Sol2 = Sol2 // Simplify;**
Sol2 // TableForm

Out[71]//TableForm=

$$\Psi'[0] \to -\frac{(\rho_1 \tau_1 \tau_2 + \rho_2 ((-1+\theta e_1+\theta e_2) \tau_1 + (\theta e_1+\theta e_2) \tau_2)^2) (\mu_{2,1}+\alpha \mu_{3,0})}{\tau_2 (\tau_1+\tau_2)^3}$$

$$\theta'[0] \to \frac{\tau_1 (-\rho_1 \tau_2^2+\rho_2 ((-1+\theta e_1+\theta e_2) \tau_1+(\theta e_1+\theta e_2) \tau_2)^2) (\alpha \mu_{2,1}-\mu_{0,2} \mu_{3,0})}{\tau_2 (\tau_1+\tau_2)^3 \mu_{0,2}}$$

$$\phi'[0] \to -\frac{\tau_1 (-\rho_1 \tau_2^2+\rho_2 ((-1+\theta e_1+\theta e_2) \tau_1+(\theta e_1+\theta e_2) \tau_2)^2) \mu_{2,1}}{\tau_2 (\tau_1+\tau_2)^3 \mu_{0,2}}$$

$$\Pi'[0] \to -\frac{(\rho_1 \tau_1 \tau_2+\rho_2 ((-1+\theta e_1+\theta e_2) \tau_1+(\theta e_1+\theta e_2) \tau_2)^2) \mu_{3,0}}{\tau_2 (\tau_1+\tau_2)^3}$$

We can further simplify it by applying the equilibrium substitutions.

In[72]:= **Sol2 = Sol2 /. EqmSubs // Simplify;**
Sol2 // TableForm

Out[73]//TableForm=

$$\Psi'[0] \to -\frac{(\rho_1 \tau_1+\rho_2 \tau_2) (\mu_{2,1}+\alpha \mu_{3,0})}{(\tau_1+\tau_2)^3}$$

$$\theta'[0] \to \frac{(\rho_1-\rho_2) \tau_1 \tau_2 (-\alpha \mu_{2,1}+\mu_{0,2} \mu_{3,0})}{(\tau_1+\tau_2)^3 \mu_{0,2}}$$

$$\phi'[0] \to \frac{(\rho_1-\rho_2) \tau_1 \tau_2 \mu_{2,1}}{(\tau_1+\tau_2)^3 \mu_{0,2}}$$

$$\Pi'[0] \to -\frac{(\rho_1 \tau_1+\rho_2 \tau_2) \mu_{3,0}}{(\tau_1+\tau_2)^3}$$

The solution for $\phi'[0]$ in Sol2 proves the assertions in Theorem 9.

If there is no derivative security, then the solution is

In[74]:= **Sol2NoY = Sol2 //. NoDeriv**

Out[74]=

$$\left\{\Psi'[0] \to -\frac{\alpha (\rho_1 \tau_1 + \rho_2 \tau_2) \mu_{3,0}}{(\tau_1 + \tau_2)^3}, \theta'[0] \to \frac{(\rho_1 - \rho_2) \tau_1 \tau_2 \mu_{3,0}}{(\tau_1 + \tau_2)^3}, \phi'[0] \to 0, \Pi'[0] \to -\frac{(\rho_1 \tau_1 + \rho_2 \tau_2) \mu_{3,0}}{(\tau_1 + \tau_2)^3}\right\}$$

This proves Theorem 8. The solution for $\theta'[0]$ in Sol2NoY proves the assertion in equation (26) and the solution for $\Pi'[0]$ in Sol2NoY proves the assertion in equation (27).

■ Compute the change in equity demand from introduction of Y

Notice that the presence of Y has no impact on the equity premium derivative, $\Pi'[0]$, but it does affect equity demand, $\theta'[0]$. So, we now determine that effect.

If there are two assets, then $\theta'[0]$ is

In[75]:= **TwoAsset = θ'[0] /. BifPt /. Sol2**

Out[75]=

$$\frac{(\rho_1 - \rho_2) \tau_1 \tau_2 (-\alpha \mu_{2,1} + \mu_{0,2} \mu_{3,0})}{(\tau_1 + \tau_2)^3 \mu_{0,2}}$$

If α=0 then the derivative asset has no impact on equilibrium and it is as if it did not exist. The substitutions in NoDeriv (defined above) state that the derivative is uncorrelated with equity at all moments, representing the case where there is no derivative. Therefore, equity demand in its absence is

```
In[76]:= SingleAsset = TwoAsset /. NoDeriv
```

$$\text{Out[76]=}\quad \frac{(\rho_1 - \rho_2)\ \tau_1\ \tau_2\ \mu_{3,0}}{(\tau_1 + \tau_2)^3}$$

The change in equity demand is the difference:

```
In[77]:= TwoAsset - SingleAsset // Simplify
```

$$\text{Out[77]=}\quad \frac{\alpha\ (-\rho_1 + \rho_2)\ \tau_1\ \tau_2\ \mu_{2,1}}{(\tau_1 + \tau_2)^3\ \mu_{0,2}}$$

This is the result reported in Theorem 10.

■ Equation 4:

We next compute (θ''[0],ϕ''[0],Ψ''[0],Π''[0]) by setting the ϵ^3 components of the equilibrium equations equal to zero.

```
In[78]:= Eq1 = Eqns11[[4]] //. UtilParams /. Sol2 /. BifPt /. {θe2 → Θ - θe1} // Expand;
         Eq2 = Eqns21[[4]] //. UtilParams /. Sol2 /. BifPt /. {θe2 → Θ - θe1} // Expand;
         Eq3 = Eqns12[[4]] //. UtilParams /. Sol2 /. BifPt /. {θe2 → Θ - θe1} // Expand;
         Eq4 = Eqns22[[4]] //. UtilParams /. Sol2 /. BifPt /. {θe2 → Θ - θe1} // Expand;
```

Here is the first equation; the rest are similar.

```
In[82]:= Eq1
```

$$\text{Out[82]=}\quad -\frac{1}{(\tau_1+\tau_2)^3} + \frac{3\,\rho_1}{(\tau_1+\tau_2)^3} + \frac{\theta e_1}{\tau_1\,(\tau_1+\tau_2)^2} - \frac{2\,\theta e_1\,\rho_1}{\tau_1\,(\tau_1+\tau_2)^2} + \frac{2\,\rho_1^2\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^4\,\mu_{0,2}} - \frac{2\,\rho_1\,\rho_2\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^4\,\mu_{0,2}} + \frac{2\,\rho_1^2\,\tau_2\,\mu_{3,0}^2}{(\tau_1+\tau_2)^4} - \frac{2\,\rho_1\,\rho_2\,\tau_2\,\mu_{3,0}^2}{(\tau_1+\tau_2)^4} + \frac{\kappa_1\,\mu_{4,0}}{2\,(\tau_1+\tau_2)^3} - \frac{\theta''[0]}{2\,\tau_1} + \frac{\Pi''[0]}{2} - \frac{\alpha\,\phi''[0]}{2\,\tau_1}$$

Solve for the second derivatives

```
In[83]:= Sol3 =
           Solve[{Eq1 == 0, Eq2 == 0, Eq3 == 0, Eq4 == 0}, {θ''[0], ϕ''[0], Ψ''[0], Π''[0]}][[1]];
```

In[84]:= **Sol3 = Sol3 // Simplify;**
Sol3

Out[85]= $\{\Psi''[0] \to -\frac{1}{(\tau_1+\tau_2)^5\,\mu_{0,2}}(2\,\alpha\,\rho_2\,(\tau_1+\tau_2)\,(2\,(-1+\theta e_1)\,\tau_1+(1+2\,\theta e_1)\,\tau_2)\,\mu_{0,2}+ 4\,\rho_1^2\,\tau_1\,\tau_2\,(\mu_{1,2}\,\mu_{2,1}+\alpha\,\mu_{2,1}^2+\mu_{0,2}\,\mu_{2,1}\,\mu_{3,0}+\alpha\,\mu_{0,2}\,\mu_{3,0}^2)+4\,\rho_2^2\,\tau_1\,\tau_2\,(\mu_{1,2}\,\mu_{2,1}+\alpha\,\mu_{2,1}^2+\mu_{0,2}\,\mu_{2,1}\,\mu_{3,0}+\alpha\,\mu_{0,2}\,\mu_{3,0}^2)-2\,\rho_1\,(\alpha\,(-3+2\,\theta e_1)\,\tau_1^2\,\mu_{0,2}+2\,\alpha\,\theta e_1\,\tau_2^2\,\mu_{0,2}+ \tau_1\,\tau_2\,(4\,\rho_2\,\mu_{2,1}\,(\mu_{1,2}+\alpha\,\mu_{2,1})+\mu_{0,2}\,(-3\,\alpha+4\,\alpha\,\theta e_1+4\,\rho_2\,\mu_{3,0}\,(\mu_{2,1}+\alpha\,\mu_{3,0}))))+ (\tau_1+\tau_2)\,(\kappa_1\,\tau_1+\kappa_2\,\tau_2)\,\mu_{0,2}\,(\mu_{3,1}+\alpha\,\mu_{4,0})),\ \theta''[0]\to\frac{1}{(\tau_1+\tau_2)^5\,\mu_{0,2}^2}$
$(-2\,(-1+\theta e_1)\,(-1+2\,\rho_2)\,\tau_1^3\,\mu_{0,2}^2+2\,\theta e_1\,(1-2\,\rho_1)\,\tau_2^3\,\mu_{0,2}^2+\tau_1^2\,\tau_2\,(4\,\alpha\,\rho_2\,(-\rho_1+\rho_2)\,\mu_{1,2}\,\mu_{2,1}+ \mu_{0,2}\,(4\,\rho_1\,\rho_2\,\mu_{2,1}\,(\mu_{2,1}-\alpha\,\mu_{3,0})-4\,\rho_2^2\,\mu_{2,1}\,(\mu_{2,1}-\alpha\,\mu_{3,0})+\alpha\,(-\kappa_1+\kappa_2)\,\mu_{3,1})+ \mu_{0,2}^2\,(-4+2\,\rho_2-2\,\theta e_1\,(-3+2\,\rho_1+4\,\rho_2)-4\,\rho_2^2\,\mu_{3,0}^2+\rho_1\,(6+4\,\rho_2\,\mu_{3,0}^2)+\kappa_1\,\mu_{4,0}-\kappa_2\,\mu_{4,0}))- \tau_1\,\tau_2^2\,(4\,\alpha\,\rho_1\,(\rho_1-\rho_2)\,\mu_{1,2}\,\mu_{2,1}+\mu_{0,2}\,(-4\,\rho_1^2\,\mu_{2,1}\,(\mu_{2,1}-\alpha\,\mu_{3,0})+ 4\,\rho_1\,\rho_2\,\mu_{2,1}\,(\mu_{2,1}-\alpha\,\mu_{3,0})+\alpha\,(\kappa_1-\kappa_2)\,\mu_{3,1})+ \mu_{0,2}^2\,(2+2\,\rho_2+2\,\theta e_1\,(-3+4\,\rho_1+2\,\rho_2)-4\,\rho_1^2\,\mu_{3,0}^2+\rho_1\,(-6+4\,\rho_2\,\mu_{3,0}^2)-\kappa_1\,\mu_{4,0}+\kappa_2\,\mu_{4,0}))),$
$\phi''[0]\to\frac{1}{(\tau_1+\tau_2)^5\,\mu_{0,2}^2}(\tau_1\,\tau_2\,(-4\,\rho_2^2\,\tau_1\,\mu_{2,1}\,(\mu_{1,2}+\mu_{0,2}\,\mu_{3,0})+4\,\rho_1\,\rho_2\,(\tau_1-\tau_2)\,\mu_{2,1} (\mu_{1,2}+\mu_{0,2}\,\mu_{3,0})+4\,\rho_1^2\,\tau_2\,\mu_{2,1}\,(\mu_{1,2}+\mu_{0,2}\,\mu_{3,0})+(\kappa_1-\kappa_2)\,(\tau_1+\tau_2)\,\mu_{0,2}\,\mu_{3,1})),$
$\Pi''[0]\to-\frac{1}{(\tau_1+\tau_2)^5\,\mu_{0,2}}(2\,\rho_2\,(\tau_1+\tau_2)\,(2\,(-1+\theta e_1)\,\tau_1+(1+2\,\theta e_1)\,\tau_2)\,\mu_{0,2}+ 4\,\rho_1^2\,\tau_1\,\tau_2\,(\mu_{2,1}^2+\mu_{0,2}\,\mu_{3,0}^2)+4\,\rho_2^2\,\tau_1\,\tau_2\,(\mu_{2,1}^2+\mu_{0,2}\,\mu_{3,0}^2)- 2\,\rho_1\,((-3+2\,\theta e_1)\,\tau_1^2\,\mu_{0,2}+2\,\theta e_1\,\tau_2^2\,\mu_{0,2}+\tau_1\,\tau_2\,(4\,\rho_2\,\mu_{2,1}^2+\mu_{0,2}\,(-3+4\,\theta e_1+4\,\rho_2\,\mu_{3,0}^2)))+ (\tau_1+\tau_2)\,(\kappa_1\,\tau_1+\kappa_2\,\tau_2)\,\mu_{0,2}\,\mu_{4,0})\}$

■ Change in equity risk premium

The presence of Y does affect Π"[0]. pi2 expresses this term.

In[86]:= **pi2 = Π''[0] /. Sol3 // Together // Expand**

Out[86]= $-\frac{6\,\rho_1\,\tau_1^2}{(\tau_1+\tau_2)^5}+\frac{4\,\theta e_1\,\rho_1\,\tau_1^2}{(\tau_1+\tau_2)^5}+\frac{4\,\rho_2\,\tau_1^2}{(\tau_1+\tau_2)^5}-\frac{4\,\theta e_1\,\rho_2\,\tau_1^2}{(\tau_1+\tau_2)^5}-\frac{6\,\rho_1\,\tau_1\,\tau_2}{(\tau_1+\tau_2)^5}+\frac{8\,\theta e_1\,\rho_1\,\tau_1\,\tau_2}{(\tau_1+\tau_2)^5}+ \frac{2\,\rho_2\,\tau_1\,\tau_2}{(\tau_1+\tau_2)^5}-\frac{8\,\theta e_1\,\rho_2\,\tau_1\,\tau_2}{(\tau_1+\tau_2)^5}+\frac{4\,\theta e_1\,\rho_1\,\tau_2^2}{(\tau_1+\tau_2)^5}-\frac{2\,\rho_2\,\tau_2^2}{(\tau_1+\tau_2)^5}-\frac{4\,\theta e_1\,\rho_2\,\tau_2^2}{(\tau_1+\tau_2)^5}-\frac{4\,\rho_1^2\,\tau_1\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^5\,\mu_{0,2}}+ \frac{8\,\rho_1\,\rho_2\,\tau_1\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^5\,\mu_{0,2}}-\frac{4\,\rho_2^2\,\tau_1\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^5\,\mu_{0,2}}-\frac{4\,\rho_1^2\,\tau_1\,\tau_2\,\mu_{3,0}^2}{(\tau_1+\tau_2)^5}+\frac{8\,\rho_1\,\rho_2\,\tau_1\,\tau_2\,\mu_{3,0}^2}{(\tau_1+\tau_2)^5}- \frac{4\,\rho_2^2\,\tau_1\,\tau_2\,\mu_{3,0}^2}{(\tau_1+\tau_2)^5}-\frac{\kappa_1\,\tau_1^2\,\mu_{4,0}}{(\tau_1+\tau_2)^5}-\frac{\kappa_1\,\tau_1\,\tau_2\,\mu_{4,0}}{(\tau_1+\tau_2)^5}-\frac{\kappa_2\,\tau_1\,\tau_2\,\mu_{4,0}}{(\tau_1+\tau_2)^5}-\frac{\kappa_2\,\tau_2^2\,\mu_{4,0}}{(\tau_1+\tau_2)^5}$

Define the price function

In[87]:= **PriceZ[ϵ_] = 1 - ϵ^2 Π[ϵ]**

Out[87]= $1-\epsilon^2\,\Pi[\epsilon]$

Compute the Taylor series for the price function

```
In[88]:= TaylorPriceZ = Normal[Series[PriceZ[ϵ] , {ϵ, 0, 4}]]
```

Out[88]= $1 - \epsilon^2\,\Pi[0] - \epsilon^3\,\Pi'[0] - \frac{1}{2}\,\epsilon^4\,\Pi''[0]$

TaylorPriceTwoAssetZ is the second-order expansion for the price of *Z* when there are two assets.

```
In[89]:= TaylorPriceTwoAssetZ =
          TaylorPriceZ //. UtilParams /. Sol3 /. Sol2 /. BifPt /. {θe₂ → Θ - θe₁} // Simplify
```

Out[89]= $1 - \frac{\epsilon^2}{\tau_1 + \tau_2} + \frac{\epsilon^3\,(\rho_1\,\tau_1 + \rho_2\,\tau_2)\,\mu_{3,0}}{(\tau_1 + \tau_2)^3} + \frac{1}{2\,(\tau_1 + \tau_2)^5\,\mu_{0,2}}$
$(\epsilon^4\,(2\,\rho_2\,(\tau_1 + \tau_2)\,(2\,(-1 + \theta e_1)\,\tau_1 + (1 + 2\,\theta e_1)\,\tau_2)\,\mu_{0,2} +$
$4\,\rho_1^2\,\tau_1\,\tau_2\,(\mu_{2,1}^2 + \mu_{0,2}\,\mu_{3,0}^2) + 4\,\rho_2^2\,\tau_1\,\tau_2\,(\mu_{2,1}^2 + \mu_{0,2}\,\mu_{3,0}^2) -$
$2\,\rho_1\,((-3 + 2\,\theta e_1)\,\tau_1^2\,\mu_{0,2} + 2\,\theta e_1\,\tau_2^2\,\mu_{0,2} + \tau_1\,\tau_2\,(4\,\rho_2\,\mu_{2,1}^2 + \mu_{0,2}\,(-3 + 4\,\theta e_1 + 4\,\rho_2\,\mu_{3,0}^2))) +$
$(\tau_1 + \tau_2)\,(\kappa_1\,\tau_1 + \kappa_2\,\tau_2)\,\mu_{0,2}\,\mu_{4,0}))$

TaylorPriceOneAssetZ is the second-order expansion for the price of Z when there is just Z. It equals TaylorPriceTwoAssetZ with the moments for *Y* and cross-moments between *Z* and *Y* zeroed out by the substitutions in NoDeriv.

```
In[90]:= TaylorPriceOneAssetZ = TaylorPriceTwoAssetZ //. NoDeriv // Simplify
```

Out[90]= $1 - \frac{\epsilon^2}{\tau_1 + \tau_2} + \frac{\epsilon^3\,(\rho_1\,\tau_1 + \rho_2\,\tau_2)\,\mu_{3,0}}{(\tau_1 + \tau_2)^3} + \frac{1}{2\,(\tau_1 + \tau_2)^5}$
$(\epsilon^4\,(2\,\rho_2\,(\tau_1 + \tau_2)\,(2\,(-1 + \theta e_1)\,\tau_1 + (1 + 2\,\theta e_1)\,\tau_2) + 4\,\rho_1^2\,\tau_1\,\tau_2\,\mu_{3,0}^2 + 4\,\rho_2^2\,\tau_1\,\tau_2\,\mu_{3,0}^2 -$
$2\,\rho_1\,((-3 + 2\,\theta e_1)\,\tau_1^2 + 2\,\theta e_1\,\tau_2^2 + \tau_1\,\tau_2\,(-3 + 4\,\theta e_1 + 4\,\rho_2\,\mu_{3,0}^2)) + (\tau_1 + \tau_2)\,(\kappa_1\,\tau_1 + \kappa_2\,\tau_2)\,\mu_{4,0}))$

We now compute the difference to determine the impact of the derivative asset on the price of *Z*.

```
In[91]:= TaylorPriceTwoAssetZ - TaylorPriceOneAssetZ // Together // Simplify
```

Out[91]= $\frac{2\,\epsilon^4\,(\rho_1 - \rho_2)^2\,\tau_1\,\tau_2\,\mu_{2,1}^2}{(\tau_1 + \tau_2)^5\,\mu_{0,2}}$

This is the result in Theorem 11.

■ Equation 5:

We next compute ($\theta'''[0]$,$\phi'''[0]$,$\Psi'''[0]$,$\Pi'''[0]$) by setting the ϵ^4 components equal to zero. We do this since these third derivatives may play a role in the utility analysis below.

```
In[92]:= Eq1 = Eqns11[[5]] //. UtilParams /. Sol3 /. Sol2 /. BifPt /. {θe₂ → Θ - θe₁};
         Eq2 = Eqns21[[5]] //. UtilParams /. Sol3 /. Sol2 /. BifPt /. {θe₂ → Θ - θe₁};
         Eq3 = Eqns12[[5]] //. UtilParams /. Sol3 /. Sol2 /. BifPt /. {θe₂ → Θ - θe₁};
         Eq4 = Eqns22[[5]] //. UtilParams /. Sol3 /. Sol2 /. BifPt /. {θe₂ → Θ - θe₁};
```

We display the first equation as an example

In[96]:= `Eq1 // Simplify`

Out[96]=

$$\frac{1}{24\,\tau_1\,(\tau_1+\tau_2)^6\,\mu_{0,2}^2}\,\Big(120\,\rho_1^3\,\tau_1\,\tau_2^2\,(\mu_{1,2}\,\mu_{2,1}^2+\mu_{0,2}\,\mu_{3,0}\,(2\,\mu_{2,1}^2+\mu_{0,2}\,\mu_{3,0}^2))+$$
$$24\,\rho_1^2\,((\tau_1+\tau_2)\,((-3+2\,\theta e_1)\,\tau_1^2-4\,(-3+\theta e_1)\,\tau_1\,\tau_2-6\,\theta e_1\,\tau_2^2)\,\mu_{0,2}^2\,\mu_{3,0}+$$
$$2\,\rho_2\,\tau_1\,(2\,\tau_1-3\,\tau_2)\,\tau_2\,(\mu_{1,2}\,\mu_{2,1}^2+\mu_{0,2}\,\mu_{3,0}\,(2\,\mu_{2,1}^2+\mu_{0,2}\,\mu_{3,0}^2)))-$$
$$12\,\rho_1\,(2\,\rho_2\,(\tau_1+\tau_2)\,(4\,(-1+\theta e_1)\,\tau_1^2+11\,\tau_1\,\tau_2-4\,\theta e_1\,\tau_2^2)\,\mu_{0,2}^2\,\mu_{3,0}+$$
$$2\,\rho_2^2\,\tau_1\,(4\,\tau_1-\tau_2)\,\tau_2\,(\mu_{1,2}\,\mu_{2,1}^2+\mu_{0,2}\,\mu_{3,0}\,(2\,\mu_{2,1}^2+\mu_{0,2}\,\mu_{3,0}^2))-\tau_2\,(\tau_1+\tau_2)\,\mu_{0,2}$$
$$(4\,\theta e_1\,\tau_2\,\mu_{0,2}\,\mu_{3,0}+\tau_1\,((5\,\kappa_1-2\,\kappa_2)\,\mu_{2,1}\,\mu_{3,1}+\mu_{0,2}\,\mu_{3,0}\,(-6+4\,\theta e_1+5\,\kappa_1\,\mu_{4,0}-2\,\kappa_2\,\mu_{4,0}))))-$$
$$(\tau_1+\tau_2)\,\mu_{0,2}\,(12\,\kappa_1\,((-4+3\,\theta e_1)\,\tau_1^2\,\mu_{0,2}\,\mu_{3,0}+3\,\theta e_1\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}+$$
$$\tau_1\,\tau_2\,(3\,\rho_2\,\mu_{2,1}\,\mu_{3,1}+\mu_{0,2}\,\mu_{3,0}\,(-4+6\,\theta e_1+3\,\rho_2\,\mu_{4,0})))+$$
$$\mu_{0,2}\,(24\,\rho_2\,\tau_2\,((-3+2\,\theta e_1)\,\tau_1+2\,\theta e_1\,\tau_2)\,\mu_{3,0}-(\tau_1+\tau_2)\,(\tau_1^5\,(\mu_{5,0}\,\mathbb{A}_1[5]+4\,\Pi^{(3)}[0])-$$
$$4\,\tau_2^4\,(\theta^{(3)}[0]+\alpha\,\phi^{(3)}[0])-4\,\tau_1^4\,(\theta^{(3)}[0]-4\,\tau_2\,\Pi^{(3)}[0]+\alpha\,\phi^{(3)}[0])+$$
$$4\,\tau_1\,\tau_2^3\,(\tau_2\,\Pi^{(3)}[0]-4\,(\theta^{(3)}[0]+\alpha\,\phi^{(3)}[0]))+8\,\tau_1^2\,\tau_2^2\,(2\,\tau_2\,\Pi^{(3)}[0]-$$
$$3\,(\theta^{(3)}[0]+\alpha\,\phi^{(3)}[0]))+8\,\tau_1^3\,\tau_2\,(3\,\tau_2\,\Pi^{(3)}[0]-2\,(\theta^{(3)}[0]+\alpha\,\phi^{(3)}[0]))))))\Big)$$

We now solve for the third derivatives. We do not display any of the results since they are very long and difficult to interpret.

In[97]:=
```
Sol4 = Solve[{Eq1 == 0, Eq2 == 0, Eq3 == 0, Eq4 == 0},
    {θ'''[0], ϕ'''[0], Ψ'''[0], Π'''[0]}][[1]];
```

Utility Expansion

We want to evaluate the impact of the derivative security Y on utility. This is necessary to derive the results in Section 6.4.

■ Utility

W_1 is the final wealth and consumption of a type 1 agent. Express it in terms of the random variables z and y, and portfolio holdings.

In[98]:= `W1 = Expand[W1]`

Out[98]= $z\,\epsilon\,\theta+\epsilon^2\,\theta\,\Pi+y\,\epsilon\,\phi+z\,\alpha\,\epsilon\,\phi+\epsilon^2\,\phi\,\Psi+Be_1+\theta e_1-\epsilon^2\,\Pi\,\theta e_1$

Define equilibrium expected utility for type 1 agents by substituting the equilibrium functions for θ, ϕ, Π, and Ψ into our expression for final consumption. The result is equilibrium utility as a function of ϵ.

In[99]:= `U1 = u1[W1] /. PortfolioSubs /. PremiaSubs`

Out[99]= $u_1[Be_1+\theta e_1+z\,\epsilon\,\theta[\epsilon]-\epsilon^2\,\theta e_1\,\Pi[\epsilon]+\epsilon^2\,\theta[\epsilon]\,\Pi[\epsilon]+y\,\epsilon\,\phi[\epsilon]+z\,\alpha\,\epsilon\,\phi[\epsilon]+\epsilon^2\,\phi[\epsilon]\,\Psi[\epsilon]]$

Compute the degree 5 Taylor series of the utility of type 1 agents in terms of ϵ and call it U1pow.

In[100]:= `U1pow = Normal[Series[U1 , {ϵ, 0, 5}]] //. UtilDerivs;`

Expand it so that products of z and y are collected

In[101]:= `U1pow = Expand[U1pow];`

Compute expected utility by replacing instances of powers of z and y with their moment expressions

```
In[102]:= U1pow = U1pow //. moments;
```

Replace θ, ϕ, Π, and Ψ and their derivatives with the solutions derived above.

```
In[103]:= U1pow = U1pow /. Sol4 /. Sol3 /. Sol2 /. BifPt //. UtilDerivs;
```

Express in terms of τ, ρ, and κ.

```
In[104]:= U1pow = U1pow //. UtilParams;
```

Combine like powers of ϵ and put the coefficients in the list U1ϵpows.

```
In[105]:= U1ϵpows = CoefficientList[U1pow, ϵ];
```

Display the coefficients of 1, ϵ, ϵ^2, and ϵ^3

```
In[106]:= Table[U1ϵpows[[i]], {i, 1, 4}] // Simplify
```

Out[106]=
$$\left\{u_1[Be_1 + \theta e_1],\ 0,\ \frac{\tau_1 - 2\,\theta e_1\,(\tau_1 + \tau_2)}{2\,(\tau_1 + \tau_2)^2},\ \frac{(3\,\rho_2\,\tau_2\,((-1 + \theta e_1)\,\tau_1 + \theta e_1\,\tau_2) + \rho_1\,\tau_1\,((-2 + 3\,\theta e_1)\,\tau_1 + (1 + 3\,\theta e_1)\,\tau_2))\,\mu_{3,0}}{3\,(\tau_1 + \tau_2)^4}\right\}$$

This list shows that the moments of Y and its cross-moments with Z have no impact on utility up to the order ϵ^3. Therefore, we move to the fourth-order terms to determine the utility impact of the derivative security.

■ Utility difference - order 4

Compute the utility contribution of the new asset. The utility difference is the utility with a nontrivial Y minus the utility with trivial Y, that is, a Y with zero co-moments with z.

Util4 is the degree four term in the utility expansion with two assets and equals

In[107]:= **Util4 = U1ϵpows[[5]] // Together**

$$\text{Out[107]=}\quad \frac{1}{8\,\tau_1\,(\tau_1+\tau_2)^6\,\mu_{0,2}}$$

$$\begin{aligned}
&(-4\,\tau_1^4\,\mu_{0,2}+8\,\theta\mathrm{e}_1\,\tau_1^4\,\mu_{0,2}-4\,\theta\mathrm{e}_1^2\,\tau_1^4\,\mu_{0,2}-16\,\rho_1\,\tau_1^4\,\mu_{0,2}+32\,\theta\mathrm{e}_1\,\rho_1\,\tau_1^4\,\mu_{0,2}-16\,\theta\mathrm{e}_1^2\,\rho_1\,\tau_1^4\,\mu_{0,2}+\\
&\quad 16\,\rho_2\,\tau_1^4\,\mu_{0,2}-32\,\theta\mathrm{e}_1\,\rho_2\,\tau_1^4\,\mu_{0,2}+16\,\theta\mathrm{e}_1^2\,\rho_2\,\tau_1^4\,\mu_{0,2}-8\,\tau_1^3\,\tau_2\,\mu_{0,2}+24\,\theta\mathrm{e}_1\,\tau_1^3\,\tau_2\,\mu_{0,2}-\\
&\quad 16\,\theta\mathrm{e}_1^2\,\tau_1^3\,\tau_2\,\mu_{0,2}-8\,\rho_1\,\tau_1^3\,\tau_2\,\mu_{0,2}+56\,\theta\mathrm{e}_1\,\rho_1\,\tau_1^3\,\tau_2\,\mu_{0,2}-48\,\theta\mathrm{e}_1^2\,\rho_1\,\tau_1^3\,\tau_2\,\mu_{0,2}+8\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}-\\
&\quad 56\,\theta\mathrm{e}_1\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}+48\,\theta\mathrm{e}_1^2\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}-4\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+24\,\theta\mathrm{e}_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}-24\,\theta\mathrm{e}_1^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+\\
&\quad 8\,\rho_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+16\,\theta\mathrm{e}_1\,\rho_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}-48\,\theta\mathrm{e}_1^2\,\rho_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}-8\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}-16\,\theta\mathrm{e}_1\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+\\
&\quad 48\,\theta\mathrm{e}_1^2\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+8\,\theta\mathrm{e}_1\,\tau_1\,\tau_2^3\,\mu_{0,2}-16\,\theta\mathrm{e}_1^2\,\tau_1\,\tau_2^3\,\mu_{0,2}-8\,\theta\mathrm{e}_1\,\rho_1\,\tau_1\,\tau_2^3\,\mu_{0,2}-16\,\theta\mathrm{e}_1^2\,\rho_1\,\tau_1\,\tau_2^3\,\mu_{0,2}+\\
&\quad 8\,\theta\mathrm{e}_1\,\rho_2\,\tau_1\,\tau_2^3\,\mu_{0,2}+16\,\theta\mathrm{e}_1^2\,\rho_2\,\tau_1\,\tau_2^3\,\mu_{0,2}-4\,\theta\mathrm{e}_1^2\,\tau_2^4\,\mu_{0,2}-16\,\rho_1^2\,\tau_1^3\,\tau_2\,\mu_{2,1}^2+16\,\theta\mathrm{e}_1\,\rho_1^2\,\tau_1^3\,\tau_2\,\mu_{2,1}^2+\\
&\quad 32\,\rho_1\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{2,1}^2-32\,\theta\mathrm{e}_1\,\rho_1\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{2,1}^2-16\,\rho_2^2\,\tau_1^3\,\tau_2\,\mu_{2,1}^2+16\,\theta\mathrm{e}_1\,\rho_2^2\,\tau_1^3\,\tau_2\,\mu_{2,1}^2+\\
&\quad 4\,\rho_1^2\,\tau_1^2\,\tau_2^2\,\mu_{2,1}^2+16\,\theta\mathrm{e}_1\,\rho_1^2\,\tau_1^2\,\tau_2^2\,\mu_{2,1}^2-8\,\rho_1\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{2,1}^2-32\,\theta\mathrm{e}_1\,\rho_1\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{2,1}^2+4\,\rho_2^2\,\tau_1^2\,\tau_2^2\,\mu_{2,1}^2+\\
&\quad 16\,\theta\mathrm{e}_1\,\rho_2^2\,\tau_1^2\,\tau_2^2\,\mu_{2,1}^2-16\,\rho_1^2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2+16\,\theta\mathrm{e}_1\,\rho_1^2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2+32\,\rho_1\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2-\\
&\quad 32\,\theta\mathrm{e}_1\,\rho_1\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2-16\,\rho_2^2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2+16\,\theta\mathrm{e}_1\,\rho_2^2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2+4\,\rho_1^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2+\\
&\quad 16\,\theta\mathrm{e}_1\,\rho_1^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2-8\,\rho_1\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2-32\,\theta\mathrm{e}_1\,\rho_1\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2+4\,\rho_2^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2+\\
&\quad 16\,\theta\mathrm{e}_1\,\rho_2^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2-3\,\kappa_1\,\tau_1^4\,\mu_{0,2}\,\mu_{4,0}+4\,\theta\mathrm{e}_1\,\kappa_1\,\tau_1^4\,\mu_{0,2}\,\mu_{4,0}-2\,\kappa_1\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{4,0}+\\
&\quad 8\,\theta\mathrm{e}_1\,\kappa_1\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{4,0}-4\,\kappa_2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{4,0}+4\,\theta\mathrm{e}_1\,\kappa_2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{4,0}+\kappa_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{4,0}+\\
&\quad 4\,\theta\mathrm{e}_1\,\kappa_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{4,0}-4\,\kappa_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{4,0}+8\,\theta\mathrm{e}_1\,\kappa_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{4,0}+4\,\theta\mathrm{e}_1\,\kappa_2\,\tau_1\,\tau_2^3\,\mu_{0,2}\,\mu_{4,0})
\end{aligned}$$

This term is very complex and difficult to interpret. However, since we are only interested in the impact of the new security on utility, we do not need to understand all of Util4. We only need to understand how the new security affects Util4. Examination of Util4 shows that $\mu_{2,1}$ is the only moment involving the new asset in the expression Util4. The degree four term in the utility expansion for the case with only one asset is computed by evaluating Util4 with $\mu_{2,1}$ set equal to zero.

In[108]:= **(Util4 /. { $\mu_{2,1}$ -> 0})**

$$\text{Out[108]=}\quad \frac{1}{8\,\tau_1\,(\tau_1+\tau_2)^6\,\mu_{0,2}}$$

$$\begin{aligned}
&(-4\,\tau_1^4\,\mu_{0,2}+8\,\theta\mathrm{e}_1\,\tau_1^4\,\mu_{0,2}-4\,\theta\mathrm{e}_1^2\,\tau_1^4\,\mu_{0,2}-16\,\rho_1\,\tau_1^4\,\mu_{0,2}+32\,\theta\mathrm{e}_1\,\rho_1\,\tau_1^4\,\mu_{0,2}-16\,\theta\mathrm{e}_1^2\,\rho_1\,\tau_1^4\,\mu_{0,2}+\\
&\quad 16\,\rho_2\,\tau_1^4\,\mu_{0,2}-32\,\theta\mathrm{e}_1\,\rho_2\,\tau_1^4\,\mu_{0,2}+16\,\theta\mathrm{e}_1^2\,\rho_2\,\tau_1^4\,\mu_{0,2}-8\,\tau_1^3\,\tau_2\,\mu_{0,2}+24\,\theta\mathrm{e}_1\,\tau_1^3\,\tau_2\,\mu_{0,2}-\\
&\quad 16\,\theta\mathrm{e}_1^2\,\tau_1^3\,\tau_2\,\mu_{0,2}-8\,\rho_1\,\tau_1^3\,\tau_2\,\mu_{0,2}+56\,\theta\mathrm{e}_1\,\rho_1\,\tau_1^3\,\tau_2\,\mu_{0,2}-48\,\theta\mathrm{e}_1^2\,\rho_1\,\tau_1^3\,\tau_2\,\mu_{0,2}+\\
&\quad 8\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}-56\,\theta\mathrm{e}_1\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}+48\,\theta\mathrm{e}_1^2\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}-4\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+24\,\theta\mathrm{e}_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}-\\
&\quad 24\,\theta\mathrm{e}_1^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+8\,\rho_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+16\,\theta\mathrm{e}_1\,\rho_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}-48\,\theta\mathrm{e}_1^2\,\rho_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}-8\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}-\\
&\quad 16\,\theta\mathrm{e}_1\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+48\,\theta\mathrm{e}_1^2\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}+8\,\theta\mathrm{e}_1\,\tau_1\,\tau_2^3\,\mu_{0,2}-16\,\theta\mathrm{e}_1^2\,\tau_1\,\tau_2^3\,\mu_{0,2}-\\
&\quad 8\,\theta\mathrm{e}_1\,\rho_1\,\tau_1\,\tau_2^3\,\mu_{0,2}-16\,\theta\mathrm{e}_1^2\,\rho_1\,\tau_1\,\tau_2^3\,\mu_{0,2}+8\,\theta\mathrm{e}_1\,\rho_2\,\tau_1\,\tau_2^3\,\mu_{0,2}+16\,\theta\mathrm{e}_1^2\,\rho_2\,\tau_1\,\tau_2^3\,\mu_{0,2}-\\
&\quad 4\,\theta\mathrm{e}_1^2\,\tau_2^4\,\mu_{0,2}-16\,\rho_1^2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2+16\,\theta\mathrm{e}_1\,\rho_1^2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2+32\,\rho_1\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2-\\
&\quad 32\,\theta\mathrm{e}_1\,\rho_1\,\rho_2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2-16\,\rho_2^2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2+16\,\theta\mathrm{e}_1\,\rho_2^2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{3,0}^2+4\,\rho_1^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2+\\
&\quad 16\,\theta\mathrm{e}_1\,\rho_1^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2-8\,\rho_1\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2-32\,\theta\mathrm{e}_1\,\rho_1\,\rho_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2+4\,\rho_2^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2+\\
&\quad 16\,\theta\mathrm{e}_1\,\rho_2^2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{3,0}^2-3\,\kappa_1\,\tau_1^4\,\mu_{0,2}\,\mu_{4,0}+4\,\theta\mathrm{e}_1\,\kappa_1\,\tau_1^4\,\mu_{0,2}\,\mu_{4,0}-2\,\kappa_1\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{4,0}+\\
&\quad 8\,\theta\mathrm{e}_1\,\kappa_1\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{4,0}-4\,\kappa_2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{4,0}+4\,\theta\mathrm{e}_1\,\kappa_2\,\tau_1^3\,\tau_2\,\mu_{0,2}\,\mu_{4,0}+\kappa_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{4,0}+\\
&\quad 4\,\theta\mathrm{e}_1\,\kappa_1\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{4,0}-4\,\kappa_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{4,0}+8\,\theta\mathrm{e}_1\,\kappa_2\,\tau_1^2\,\tau_2^2\,\mu_{0,2}\,\mu_{4,0}+4\,\theta\mathrm{e}_1\,\kappa_2\,\tau_1\,\tau_2^3\,\mu_{0,2}\,\mu_{4,0})
\end{aligned}$$

We take the difference in the past two expressions to compute the impact of the new security.

```
In[109]:= UtilDiff = (Util4 - (Util4 /. { μ2,1 -> 0})) // Expand
```

$$\text{Out[109]=}\ -\frac{2\,\rho_1^2\,\tau_1^2\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}+\frac{2\,\theta e_1\,\rho_1^2\,\tau_1^2\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}+\frac{4\,\rho_1\,\rho_2\,\tau_1^2\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}-\frac{4\,\theta e_1\,\rho_1\,\rho_2\,\tau_1^2\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}-\frac{2\,\rho_2^2\,\tau_1^2\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}+\frac{2\,\theta e_1\,\rho_2^2\,\tau_1^2\,\tau_2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}+\frac{\rho_1^2\,\tau_1\,\tau_2^2\,\mu_{2,1}^2}{2\,(\tau_1+\tau_2)^6\,\mu_{0,2}}+\frac{2\,\theta e_1\,\rho_1^2\,\tau_1\,\tau_2^2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}-\frac{\rho_1\,\rho_2\,\tau_1\,\tau_2^2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}-\frac{4\,\theta e_1\,\rho_1\,\rho_2\,\tau_1\,\tau_2^2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}+\frac{\rho_2^2\,\tau_1\,\tau_2^2\,\mu_{2,1}^2}{2\,(\tau_1+\tau_2)^6\,\mu_{0,2}}+\frac{2\,\theta e_1\,\rho_2^2\,\tau_1\,\tau_2^2\,\mu_{2,1}^2}{(\tau_1+\tau_2)^6\,\mu_{0,2}}$$

This is a much simpler expression. We simplify the difference to arrive at the result in the paper.

```
In[110]:= UtilDiff = Simplify[UtilDiff]
```

$$\text{Out[110]=}\ \frac{(\rho_1-\rho_2)^2\,\tau_1\,\tau_2\,(4\,(-1+\theta e_1)\,\tau_1+(1+4\,\theta e_1)\,\tau_2)\,\mu_{2,1}^2}{2\,(\tau_1+\tau_2)^6\,\mu_{0,2}}$$

Use the identity $-1+\theta e_1-\theta e_2$:

```
In[111]:= UtilDiff = UtilDiff /. {-1 + θe1 → -θe2}
```

$$\text{Out[111]=}\ \frac{(\rho_1-\rho_2)^2\,\tau_1\,\tau_2\,(-4\,\theta e_2\,\tau_1+(1+4\,\theta e_1)\,\tau_2)\,\mu_{2,1}^2}{2\,(\tau_1+\tau_2)^6\,\mu_{0,2}}$$

The expression in Theorem 12 is

$$\text{In[112]:= Thm12} = \frac{(\rho_1-\rho_2)^2\,\tau_1{}^2\,\tau_2{}^2}{2\,(\tau_1+\tau_2)^6}\left(4\left(\frac{\theta e_1}{\tau_1}-\frac{\theta e_2}{\tau_2}\right)+\frac{1}{\tau_1}\right)\frac{\mu_{2,1}^2}{\mu_{0,2}}$$

$$\text{Out[112]=}\ \frac{(\rho_1-\rho_2)^2\,\tau_1^2\left(\frac{1}{\tau_1}+4\left(\frac{\theta e_1}{\tau_1}-\frac{\theta e_2}{\tau_2}\right)\right)\tau_2^2\,\mu_{2,1}^2}{2\,(\tau_1+\tau_2)^6\,\mu_{0,2}}$$

```
In[113]:= UtilDiff - Thm12 // Simplify
Out[113]= 0
```

Therefore, our UtilDiff expression is equivalent to the one for utility change in Theorem 12.